STATIC VERSION

module routing_static

sig IP {}

sig Link {from, to: Router}

sig Router {ip: IP, table: IP ->? Link, nexts: set Router}


{


table[IP].from in this


nexts = table[IP].to


no table[ip]


}

fact {inj (Router$ip)}

fun inj [t,t'] (r:t->t') {all x: t' | sole r.x} 

fun Consistent () {


all r: Router, i: IP |



r.table[i].to in i.~ip.*~nexts


}

LOCAL STATE MACHINE

sig State {}

sig IP {}

sig Link {from, to: State ->! Router}

sig Router {ip: IP, table: State -> IP ->? Link, nexts: State -> Router}


{


all s: State {



(table[s][IP].from)[s] in this



nexts[s] = (table[s][IP].to)[s]



no table[s][ip]}


}

fact {inj (Router$ip)}

fun Consistent (s: State) {


let rnexts = {r,r': Router | r->s->r' in Router$nexts} |


all r: Router, i: IP |



(r.table[s][i].to)[s] in i.~ip.*~rnexts


}

fun Propagate (s, s': State) {


let rnexts = {r,r': Router | r->s->r' in Router$nexts} |


all r: Router |



r.table[s'] in r.table[s] + r.~rnexts.table[s]


}

assert PropagationOK {


all s, s': State |



Consistent (s) && Propagate (s,s') => Consistent (s')


}

GLOBAL STATE MACHINE

sig IP {}

sig Link {}

sig Router {}

sig LinkState {


from, to: Link -> Router


}

sig NetworkState extends LinkState {


ip: Router -> IP,


table: Router -> IP ->? Link,


nexts: Router -> Router }


{


all r: Router {



table[r][IP].from in r



nexts[r] = table[r][IP].to



no table[r][r.ip]



}


inj (ip)


}

fun Consistent (s: NetworkState) {


all r: Router, i: IP |



s.table[r][i].to[s] in i.~(ip[s]).*~(nexts[s])


}

fun Propagate (s, s': NetworkState) {


all r: Router {



s'.table[r] in s.table[r] + r.~(nexts[s]).(table[s])



s.ip[r] = s'.ip[r]



}


}

assert PropagationOK {


all s, s': NetworkState |



Consistent (s) && Propagate (s,s') => Consistent (s')


}

OBJECT ORIENTED

sig IP {}

sig Link {from, to: RouterRef}

sig Router {ip: IP, table: IP ->? LinkRef, nexts: set RouterRef}


{


no table[ip]


}

fact {inj (Router$ip)}

fun inj [t,t'] (r:t->t') {all x: t' | sole r.x} 

sig LinkRef {}

sig RouterRef {}

sig State {


robj: RouterRef ->! Router,


lobj: LinkRef ->! Link


}

fact {


all s: State, r: RouterRef {



r.(s.robj).table[IP].(s.lobj).from in r



r.(s.robj).nexts = r.(s.robj).table[IP].(s.lobj).to



}


}

fun Consistent (s: State) {


all r: RouterRef, i: IP |



r.(s.robj).table[i].(s.lobj).to in i.~ip.~(s.robj).*~(s.robj.nexts)


}

fun Propagate (s, s': State) {


let rnexts = {r,r': Router | r->s->r' in Router$nexts} |


all r: RouterRef |



r.(s'.robj).table in r.(s.robj).table




+ r.~(s.robj.nexts).(s.robj).table


}

assert PropagationOK {


all s, s': State |



Consistent (s) && Propagate (s,s') => Consistent (s')


}

